Homework solutions

Find the adjoint of this nonlinear statement

$$x = Ax + y + z^2$$

Solution

Non-linear statement:

$$z = z$$
$$y = y$$
$$x = Ax + y + z^2$$

Tangent linear statement:

$$\delta z = \delta z$$
$$\delta y = \delta y$$
$$\delta x = A\delta x + \delta y + 2z\delta z$$

Tangent linear statement in matrix form:

$$\begin{pmatrix} \delta z \\ \delta y \\ \delta x \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 2z & 1 & A \end{pmatrix} \begin{pmatrix} \delta z \\ \delta y \\ \delta x \end{pmatrix}$$

Adjoint statement in matrix form (note the matrix transposition):

$$\begin{pmatrix} \delta z^* \\ \delta y^* \\ \delta x^* \end{pmatrix} = \begin{pmatrix} 1 & 0 & 2z \\ 0 & 1 & 1 \\ 0 & 0 & A \end{pmatrix} \begin{pmatrix} \delta z^* \\ \delta y^* \\ \delta x^* \end{pmatrix}$$

Adjoint statement:

$$\delta z^* = \delta z^* + 2z\delta x^*$$
$$\delta y^* = \delta y^* + \delta x^*$$
$$\delta x^* = A\delta x^*$$

The tangent linear code (1)

```
SUBROUTINE model_tl(x,xd,dt,nstep)
REAL,INTENT(INOUT) :: x(3),xd(3)
REAL,INTENT(IN)    :: dt          ! constant
INTEGER, INTENT(IN):: nstep
REAL               :: y(3), yd(3)
  DO i = 1,nstep
    CALL lorenz   (x,y)        ! trajectory
    CALL lorenz_tl(x,xd,yd)
    CALL step     (x,y,dt)     ! trajectory
    CALL step_tl  (xd,yd,dt)
  ENDDO
END SUBROUTINE model_tl
```

The tangent linear code (2)

```
! ------------------------------
SUBROUTINE lorenz_tl(x,xd,yd)
REAL,INTENT(IN) :: x(3),xd(3)
REAL,INTENT(OUT):: yd(3)
REAL :: p, r, b           !      constants
  yd(1) = -p*xd(1)+p*xd(2)
  yd(2) = xd(1)*(r-x(3))-xd(2)-x(1)*xd(3)
  yd(3) = xd(1)*x(2)+x(1)*xd(2)-b*xd(3)
END SUBROUTINE lorenz_tl
```

```
! ----------------------------
SUBROUTINE step_tl (xd,yd,dt)
REAL,INTENT(INOUT):: xd(3)
REAL,INTENT(IN)   :: yd(3)
REAL,INTENT(IN)   :: dt           ! constant
  DO i = 1,3
    xd(i)=xd(i)+dt*yd(i)
  ENDDO
END SUBROUTINE step_tl
```

The adjoint code (1)

```
SUBROUTINE model_ad(x,xb,dt,nstep)
REAL,INTENT(INOUT) :: x(3),xb(3)
REAL,INTENT(IN)    :: dt    ! constant
INTEGER, INTENT(IN):: nstep
REAL               :: yb(3)

! Trajectory computations
  DO i = 1,nstep
    CALL lorenz(x,y)
    x_store(i) = x(i)      ! Store nonlinear trajectory
    CALL step(x,y,dt)
  ENDDO
! Adjoint computation
  yb(:) = 0.0
  DO i = nstep,1,-1
    call step_ad  (xb,yb,dt)
    call lorenz_ad(x_store,xb,yb)
  ENDDO
END SUBROUTINE model_ad
```

The adjoint code (2)

```
! ---------------------------------
SUBROUTINE step_ad(xb,yb,dt)
REAL,INTENT(INOUT):: xb(3),yb(3)
REAL,INTENT(IN)   :: dt            ! constant
  DO i = 3,1,-1
    yb(i) = yb(i)+dt*xb(i)
  ENDDO
END SUBROUTINE step_ad
```

```
! ---------------------------------
SUBROUTINE lorenz_ad(x,xb,yb)
REAl,INTENT(IN)    :: x(3)
REAL, INTENT(INOUT):: yb(3),xb(3)
REAL:: p, r, b                      ! constants
 xb(1) = xb(1) + x(2)*yb(3)
 xb(2) = xb(2) + x(1)*yb(3)
 xb(3) = xb(3) - b*yb(3)
 yb(3) = 0.0
 xb(1) = xb(1) + (r-x(3))*yb(2)
 xb(2) = xb(2) - yb(2)
 xb(3) = xb(3) - x(1)*yb(2)
 yb(2) = 0.0
 xb(1) = xb(1) - p*yb(1)
 xb(2) = xb(2) + p*yb(1)
 yb(1) = 0.0
END SUBROUTINE lorenz_ad
```